Pensieve Header: A first attempt at the "infinitesimal Alexander module"; continued as 2009-06/InfinitesimalAlexanderModules.nb.

```
<< KnotTheory`
```

```
Loading KnotTheory` version of August 13, 2008, 14:31:13.4448.
Read more at http://katlas.org/wiki/KnotTheory.
```

```
(# → Alexander[#][t]) & /@ AllKnots[{3, 7}]
```

KnotTheory::loading : Loading precomputed data in PD4Knots`.

$$\left\{\text{Knot}[3,1]\to -1+\frac{1}{t}+t,\ \text{Knot}[4,1]\to 3-\frac{1}{t}-t,\ \text{Knot}[5,1]\to 1+\frac{1}{t^2}-\frac{1}{t}-t+t^2,\right.$$
$$\text{Knot}[5,2]\to -3+\frac{2}{t}+2t,\ \text{Knot}[6,1]\to 5-\frac{2}{t}-2t,\ \text{Knot}[6,2]\to -3-\frac{1}{t^2}+\frac{3}{t}+3t-t^2,$$
$$\text{Knot}[6,3]\to 5+\frac{1}{t^2}-\frac{3}{t}-3t+t^2,\ \text{Knot}[7,1]\to -1+\frac{1}{t^3}-\frac{1}{t^2}+\frac{1}{t}+t-t^2+t^3,$$
$$\text{Knot}[7,2]\to -5+\frac{3}{t}+3t,\ \text{Knot}[7,3]\to 3+\frac{2}{t^2}-\frac{3}{t}-3t+2t^2,$$
$$\text{Knot}[7,4]\to -7+\frac{4}{t}+4t,\ \text{Knot}[7,5]\to 5+\frac{2}{t^2}-\frac{4}{t}-4t+2t^2,$$
$$\left.\text{Knot}[7,6]\to -7-\frac{1}{t^2}+\frac{5}{t}+5t-t^2,\ \text{Knot}[7,7]\to 9+\frac{1}{t^2}-\frac{5}{t}-5t+t^2\right\}$$

```
K = Knot[7, 7];
pd = PD[K];
gc = GC @@ pd /. X[i_, j_, k_, l_] :> If[PositiveQ[X[i, j, k, l]],
      Ar[l, i, +1], Ar[j, i, -1]
     ];
(* gc = GC[Ar[1,3, +1], Ar[4,2,-1]]; *)
n = 2 Length[gc];
gc
```

```
GC[Ar[4, 1, -1], Ar[10, 5, -1], Ar[8, 3, 1],
 Ar[2, 9, 1], Ar[14, 11, -1], Ar[12, 7, 1], Ar[6, 13, 1]]
```

Conventions for red objects:

1. Legs start just to the right of the index; RedAr[0,7] means a red arrow starting to the right of position 0 (that is, to the left of everything) and ending to the right of position 7).
2. If two (red) indices are the same, the heads are to the right of the tails.
3. RedW[] is the legless wheel object.
3. RedY[i,j,k] means "red Y with tails are i and j and head at k".

```
range = Range[0, n];
AllRedObjects = Flatten[{
    Outer[RedAr, range, range], Outer[RedY, range, range, range], RedW[]
   }];
Short[AllRedObjects]
```

```
{RedAr[0, 0], RedAr[0, 1], RedAr[0, 2], RedAr[0, 3], RedAr[0, 4],
 <<3592>>, RedY[14, 14, 12], RedY[14, 14, 13], RedY[14, 14, 14], RedW[]}
```

The relations associated with a red objects involve all the ways of "pulling one leg to the left" :

```
RelationsIn[gc_GC, red_] := ReplaceList[
  red * (Times @@ Select[gc, (Intersection[List @@ #, List @@ red] =!= {}) &]), {
   (* Short red arrows are central: *)
   RedAr[i_, i_] * _. /; i > 0 :→ -red + RedAr[i - 1, i - 1],
   (* Tails commute for RedAr: *)
   RedAr[i_, j_] Ar[i_, _, _] * _. :→ -red + RedAr[i - 1, j] + If[i - 1 == j, -x RedW[], 0],
   (* Commuting a RedAr tail across an Ar head *)
   RedAr[i_, j_] Ar[k_, i_, s_] * _. :→
    -red + RedAr[i - 1, j] + (X^s - 1) / x RedY[k, i, j] + If[i - 1 == j, -x RedW[], 0],
   (* Commuting the head of a RedAr with the head of an Ar *)
   RedAr[i_, j_] Ar[k_, j_, s_] * _. :→
    -red + RedAr[i, j - 1] - (X^(+s) - 1) / x RedY[k, i, j - 1] + If[i == j, x RedW[], 0],
   (* The anti-symmetry of RedY: *)
   RedY[i_, j_, k_] * _. :→ red + RedY[j, i, k],
   (* Tails commute for RedY: *)
   RedY[i_, j_, k_] Ar[i_, _, _] * _. :→
    -red + RedY[i - 1, j, k] + If[i - 1 == k, x^2 RedW[], 0],
   RedY[i_, j_, k_] Ar[j_, _, _] * _. :→
    -red + RedY[i, j - 1, k] + If[j - 1 == k, -x^2 RedW[], 0],
   (* Commuting a RedY tail across an Ar head *)
   RedY[i_, j_, k_] Ar[l_, j_, s_] * _. :→
    -red + RedY[i, j - 1, k] - (X^s - 1) RedY[l, j, k] + If[j - 1 == k, -x^2 RedW[], 0]
  }
 ]
```

```
AllRedRelations = Flatten[RelationsIn[gc, #] & /@ AllRedObjects];
rule = Thread[Rule[AllRedObjects, IdentityMatrix[Length[AllRedObjects]]]];
Short[RedRules = Map[
   (
     p = First@Part[AllRedObjects, First@Position[#, 1, {1}]];
     p → p - (#.AllRedObjects)
    ) &,
   DeleteCases[mat = Simplify[RowReduce[AllRedRelations /. rule]], {0 ...}]
  ]]
```

$$\left\{\text{RedAr}[0,0]\to\text{RedAr}[14,14],\ \text{RedAr}[0,1]\to\text{RedAr}[14,14]-x\left(-1+\frac{1}{X}\right)\text{RedW}[],\right.$$
$$\left.\ll 3595\gg,\ \text{RedY}[14,14,13]\to 0,\ \text{RedY}[14,14,14]\to 0\right\}$$

```
Simplify[RedRules[[Table[Random[Integer, {1, Length[RedRules]}], {10}]]]]
```

$$\Big\{\mathrm{RedY}[1, 11, 6] \to \frac{x^2\,\left(1 - 4\,X + 5\,X^2 - 4\,X^3 + X^4\right)\,\mathrm{RedW}[\,]}{X\,\left(1 - 5\,X + 9\,X^2 - 5\,X^3 + X^4\right)},\ \mathrm{RedY}[6, 2, 4] \to \frac{x^2\,X^3\,\mathrm{RedW}[\,]}{1 - 5\,X + 9\,X^2 - 5\,X^3 + X^4},$$

$$\mathrm{RedY}[3, 13, 6] \to -\frac{x^2\,\left(-2 + 7\,X - 5\,X^2 + X^3\right)\,\mathrm{RedW}[\,]}{X\,\left(1 - 5\,X + 9\,X^2 - 5\,X^3 + X^4\right)},\ \mathrm{RedY}[11, 8, 8] \to \frac{x^2\,X\,\mathrm{RedW}[\,]}{1 - 5\,X + 9\,X^2 - 5\,X^3 + X^4},$$

$$\mathrm{RedY}[9, 4, 13] \to \frac{x^2\,(-2 + X)\,(-1 + X)^2\,\mathrm{RedW}[\,]}{1 - 5\,X + 9\,X^2 - 5\,X^3 + X^4},\ \mathrm{RedY}[3, 1, 5] \to -\frac{x^2\,\left(-1 + 4\,X - 4\,X^2 + X^3\right)\,\mathrm{RedW}[\,]}{1 - 5\,X + 9\,X^2 - 5\,X^3 + X^4},$$

$$\mathrm{RedAr}[11, 2] \to \mathrm{RedAr}[14, 14] - \frac{x\,\left(2 - 10\,X + 16\,X^2 - 9\,X^3 + 2\,X^4\right)\,\mathrm{RedW}[\,]}{1 - 5\,X + 9\,X^2 - 5\,X^3 + X^4},$$

$$\mathrm{RedAr}[8, 5] \to \mathrm{RedAr}[14, 14] - \left(x + \frac{x\,(-1 + X)^2\,X}{1 - 5\,X + 9\,X^2 - 5\,X^3 + X^4}\right)\,\mathrm{RedW}[\,],$$

$$\mathrm{RedY}[1, 12, 3] \to \frac{x^2\,(-2 + X)\,X^3\,\mathrm{RedW}[\,]}{1 - 5\,X + 9\,X^2 - 5\,X^3 + X^4},\ \mathrm{RedY}[7, 8, 4] \to 0\Big\}$$

```
RedZ = Plus @@ gc /. Ar[i_, j_, s_] :> s * RedAr[i, j]
```

$$\mathrm{RedAr}[2, 9] - \mathrm{RedAr}[4, 1] + \mathrm{RedAr}[6, 13] + \mathrm{RedAr}[8, 3] - \mathrm{RedAr}[10, 5] + \mathrm{RedAr}[12, 7] - \mathrm{RedAr}[14, 11]$$

```
Simplify[RedZ /. RedRules]
```

$$\frac{\left(1 - 5\,X + 9\,X^2 - 5\,X^3 + X^4\right)\,\mathrm{RedAr}[14, 14] - x\,\left(-3 + 10\,X - 9\,X^2 + X^4\right)\,\mathrm{RedW}[\,]}{1 - 5\,X + 9\,X^2 - 5\,X^3 + X^4}$$

```
{
  RedAr[0, 0] - RedAr[1, 1],
  RedY[2, 3, 4] + RedY[3, 2, 4],
  RedY[2, 1, 0] - RedY[2, 0, 0],
  RedY[4, 2, 4] - RedY[4, 1, 4],
  RedY[4, 2, 4]
 } /. RedRules
```

$$\left\{0,\ 0,\ -x^2\,\mathrm{RedW}[\,],\ 0,\ -\frac{x^2\,X\,\left(-1 + 4\,X - 4\,X^2 + X^3\right)\,\mathrm{RedW}[\,]}{1 - 5\,X + 9\,X^2 - 5\,X^3 + X^4}\right\}$$

```
Union[Cases[Last /@ RedRules, _RedAr | _RedY | _RedW, Infinity]]
```

$$\{\mathrm{RedAr}[14, 14],\ \mathrm{RedW}[\,]\}$$